Plain BASIC Talk

by Ken Lill

A remake of Scott Waldinger's "Startrek"

The reason I'm writing this article is to show Bally/Astro BASIC programmers, new and old, some tips for speeding up programs and some space saving.

I chose Startrek not because it had mistakes, but because of the way it was written, it used almost every byte of available memory and it could use a little tweeking.

The 1st thing I did was to move all of the "setup" data to the end of the program. It is only accessed during startup. If you have a program that uses hand controls for moving items or shooting you will want to keep the motion area close to the beginning (top) of the program. This is because any GOSUB or GOTO commands have to look through the program from the 1st byte to the actual line number to start working there. The more lines it has to go through, the longer it will take.

Line 1 is simple, it tells the program to go to the setup area and start from there.

Line 2 is a simple time delay loop. In the original program (I'll call this Ver.1) he had 8 such loops. Most were the same length, so I took the average length and made it a subroutine. By putting it so close to the top, it won't affect it's speed. I could have even put it on the bottom because it is just used to keep the text and graphics on the screen long enough to see it.

Line 3 is another loop, but this one just puts data into the sound port and then in line 4 it turns it off.

Line 5 actually starts the "play" area of the program. Note that there is a GOSUB 8. This gets any of it's 5 main commands from you and starts that part of the program.

Line 6 sets up the background (BC) and foreground (FC) colors and checks to see if there are any Klingons (K variable) left. If not, end the program in line 80.

Line 7 is to restart the "play" part.

Line 8 asks for your input, from 1-5. If you enter any other number, it will redo the input command until you get it right,

Line 9 is a complete redo of Ver.1 lines150 - 175. Because I repositioned all of the routines in a certain order, and started them all with numbers divisible by 10, I was able to do it all in a simple math equation instead of the 4 "IF" statements. (#4 was a drop through starting at line 185)

From 10 through 56 are the different user command subroutines

You'll notice that I put as may commands as I could into most lines. I could have made some longer as the line input buffer is over 100 bytes long, but typing in lines that long leaves more chance for error.

Wherever I could, I used math instead of IF statements. It works faster and is cleaner. One example is in Ver.2 line 21. F=(V=X). This was ver.1 lines335, 340 and 350.

Remember that each line not only consists of the program data but also 2 bytes for the line number and 1 for the "GO".

Another great byte savings is with Ver.1 lines 623 - 680. That area is replaced in Ver.2 by 50 -56.

In line 74 K=RND (4) could be replaced with K=(RND (2)-1)xRND (3) to be acurate to what the original percentage each number will come up. (zero=50%, 1-2 or 3 are 16 2/3% each). Actually line 74, wih the exception of the CLEAR command, is lines 100 – 130 in Ver.1.

One other tip is: if a quote ends a line, you don't need an end quotation mark.

Check the byte count difference!

If you have any questions about any of the 5 main BASIC programs for the Bally / Astrocade including Hot Rod Bally BASIC, Vipersoft BASIC and Blue Ram BASIC, feel free to email me: *kenzre@yahoo.com*

Make sure that you include in the subject line "A question about a Bally BASIC". That way I will respond ASAP!

Happy Programming!

STARTREK

by SCOTT Waldinger

```
5 S=0;F=0;E=300;&(19)=20;&(20)=10                                   +34 bytes
10 clear ;NT=3;BC=16;FC=7                                           +20=54
15 for A=21to 40;@(A)=50;next A                                     +22=76
25 A=0                                                              + 6=82
30 clear ;gosub 100                                                 + 9=91
35 CY=10;print ;print #3,A,"KLINGONS"                               +27=118
36 K=A                                                              + 6=124
37 for A=1to 800;next A                                             +14=138
38 clear                                                            + 4=142
40 print "QUAD",#3,Q," SEC",#3,T                                   +27=169
50 gosub 140                                                        + 7=176
60 clear ;BC=16;FC=7                                                +15=191
65 if K<1goto 700                                                   +11=202
70 goto 40                                                          + 6=208
100 for I=1to 20                                                    +10=218
105 P=rnd (2)-1                                                     +11=229
106 if P=1P=rnd (3)                                                 +13=242
108 A=A+P                                                           + 8=250
110 @(I)=P                                                          + 9=259
120 next I                                                          + 5=264
130 Q=rnd (20);T=rnd (5);return                                     +19=283
140 input "COMMAND:"C                                               +15=298
150 if C=1goto 260                                                  +11=309
160 if C=2goto 330                                                  +11=320
170 if C=3goto 580                                                  +11=331
175 if C=5goto 620                                                  +11=342
185 clear ;CY=0;print "COMPUTER ACTIVE"                             +28=370
190 if @(Q)#0clear ;goto 205                                        +16=386
197 print "SENSORS:NEGATIVE"                                        +22=408
198 for A=1to 700;next A                                            +14=422
200 return                                                          + 4=426
205 BC=99;FC=0;CX=-20;print "RED ALERT"                             +32=458
220 for A=1to 800;next A                                            +14=472
225 NT=3                                                            + 7=479
230 clear                                                           + 4=483
240 CY=20;print ;Z=@(Q);print #3,Z," KLINGONS IN THIS QUADRANT"     +52=535
245 if S<10print "SHIELDS TOO LOW"                                  +26=560
247 V=rnd (10);print "TRAJ. IS ",#1,V                               +28=588
248 for A=1to 900;next A                                            +13=602
250 return                                                          + 4=606
260 clear ;CY=10;input "TO WHICH QUADRANT?"C                        +33=639
270 W=ABS(Q-C)                                                      +13=652
280 if W>6W=6                                                       +10=662
305 E=E-rnd (5)mul W                                                +13=675*
320 return                                                          + 4=679
330 if @(Q)=0return                                                 +11=690
331 clear ;input "TRAJ."X                                           +14=704
335 F=1                                                             + 6=710
340 if X=Vgoto 360                                                  +11=721
350 F=0                                                             + 6=727
760 clear ;print "PHASERS FIRED"                                    +21=748
370 clear                                                           + 4=752
375 E=E-5+rnd (10)                                                  +14=766
380 BC=0;FC=7                                                       +12=778
385 for A=1to 45;box rnd (160)-80,rnd (87)-44,1,1,1;next A          +39=817
390 X=rnd (30)-60;Y=rnd (60)-30;A=10;gosub 395;goto 410             +39=856
395 box X,Y,8,8,3;box X-A,Y,12,4,3;line X-A,Y,4;line X-A,Y+6,1      +47=903
400 line X-A,Y,4;line X-A,Y-6,1                                     +22=925
405 box (X-A)-Adiv 5+1,Y+6,8,3,1;box (X-A)-Adiv 5+1,Y-6,8,3,1      +48=973*
```

```
407 return                                                                    + 4=977
410 U=X                                                                       + 6=983
415 X=ABS(X);A=-10;gosub 395                                                 +23=1006
420 FC=126                                                                    + 9=1015
430 if F=0line -X,Y-rnd (15),1;for A=1to 200;&(21)=15;next A;&(21)=0;goto 450+54=1069
440 line -X,Y,1;for A=1to 200;&(21)=15;next A;&(21)=0                        +39=1108
441 @(Q+20)=@(Q+20)-3mul rnd (12);if @(Q+20)>0goto 450                        +40=1148*
445 &(23)=255;&(21)=255;for A=1to 800;next A                                 +34=1182
447 print "KLINGON DESTROYED";K=K-1                                           +29=1211
448 @(Q)=@(Q)-1;print #2,K," LEFT"                                           +28=1239
449 V=0;@(Q+20)=50;&(21)=0;goto 248                                          +30=1269
450 J=rnd (10)                                                               +10=1279
460 if J=1goto 500                                                           +11=1290
470 J=Jmul rnd (2)                                                           +11=1301*
480 S=S-J                                                                    + 8=1309
493 line U,Y,4;line X,Y,1                                                    +16=1325
495 &(21)=15                                                                 +11=1336
497 for A=1to 700;next A                                                     +14=1350
498 &(21)=0;goto 510                                                         +15=1365
500 line U,Y,4;line X,Y+rnd (15),1;print "MISS";goto 220                     +35=1400
510 print "HIT"                                                             + 9=1409
520 for A=1to 700;next A                                                     +14=1423
525 if S<1print "DESTROYED";goto 700                                         +24=1447
530 print "DAMAGE:",#1,S                                                     +18=1465
531 print #10,@(Q+20)                                                        +15=1480
534 for A=1to 900;next A                                                     +14=1494
535 return                                                                   + 4=1498
580 input "SHIELDS?"S                                                        +15=1513
590 if S>Egoto 580                                                           +11=1524
610 return                                                                   + 4=1528
620 clear ;print "LR SCAN"                                                   +15=1543
621 M=3                                                                      + 6=1549
623 if Q-M<1goto 660                                                         +13=1562
624 if Q-M>20goto 660                                                        +14=1576
625 print Q-M,#1,":",#1                                                      +17=1593
630 if @(Q-M)=0print "0",;goto 660                                           +22=1615
635 if @(Q-M)>9print #1,@(Q-M),;goto 660                                     +28=1643
640 print #1,@(Q-M)                                                          +13=1656
660 print                                                                    + 4=1660
670 M=M-1                                                                    + 8=1668
675 if M<-3goto 248                                                          +12=1680
680 goto 623                                                                 + 7=1687
700 clear                                                                    + 4=1691

* mul and div count as one byte.
                                                                              SZ=111
40 @( ) array positions = 80 bytes                                           +80=1771

That leaves 29 bytes of memory left.
```

```
1 goto 70                                                                    6 bytes
2 for A=1to 800;next A;return                                                +16=22
3 for A=1to 200;&(21)=15;next A                                              +23=45
4 &(21)=0;return                                                             +12=57
5 clear;print "QUAD",#4,Q,"SEC",#3,T;gosub 8                                 +31=88
6 BC=16;FC=7;if K<1goto 80                                                   +21=109
7 goto 5                                                                     +  5=114
8 input "COMMAND:"C;if (C>5)+(C=0)CY=CY+8;gosub3;goto 8                      +41=155
9 goto Cx10                                                                  + 8=163

10 clear ;CY=10;input "to WHICH QUADRANT?"C;W=ABS(Q-C);if W>6W=6             +50=213
15 E=E-rnd (5)xW;return                                                      +15=228

20 if @(Q)=0return                                                           +11=239
21 clear ;input "TRAJ."X;F=(V=X);clear ;print "PHASERS FIRED";clear ;E=E-5+rnd (10)  +55=293
22 BC=0;FC=7;for A=1to 45;box rnd (160)-80,rnd (87)-44,1,1,1;next A           +49=340
23 X=rnd (30)-60;Y=rnd (60)-30;A=10;gosub 24;goto 26                          +37=377
24 B=X-A;box X,Y,8,8,3;box B,Y,12,4,3;line B,Y,4;line B,Y+6,1                +47=424
25 line B,Y,4;line B,Y-6,1;box (B)-A÷5+1,Y+6,8,3,1;box (B)-A÷5+1,Y-6,8,3,1;return  +62=486
26 U=X;X=ABS(X);A=-10;gosub 24;FC=126;if F=0line -X,Y-rnd (15),1;gosub 3;goto 60  +57=543
27 line -X,Y,1;gosub 3;D=Q+20;@(D)=@(D)-3xrnd (12);if @(D)>0goto 60          +49=592
28 &(23)=255;&(21)=255;gosub 2;print "KLINGON DESTROYED";K=K-1               +52=644
29 @(Q)=@(Q)-1;print #3,K,"LEFT";V=0;@(D)=50;gosub 4;goto 2                   +45=689

30 input "SHIELDS?"S;if S>Egoto 30                                           +23=712
35 return                                                                    +  4=716

40 clear ;CY=0;print "COMPUTER ACTIVE                                        +27=740
42 if @(Q)=0print "SENSORS:NEGATIVE";goto 2                                  +32=772
44 BC=99;FC=0;CX=-20;print "RED ALERT                                        +32=804
46 gosub 2;NT=3;clear ;CY=12;Z=@(Q)                                          +25=829
48 print #4,Z,"KLINGONS IN THIS QUADRANT";if S<10print "SHIELDS TOO LOW       +59=888
49 V=rnd (10);print "TRAJ. IS",#2,V;goto 2                                    +30=918

50 clear ;print "LR SCAN";M=3;Z=Q-M                                          +25=943
52 if (Z<1)+(Z>20)=0print Z,#1,":",#1,@(Z)                                   +35=978
54 M=M-1;if M<-3goto 2                                                       +16=994
56 goto 52                                                                   +  6=1000

60 J=rnd (10);if J#1line U,Y,4;line X,Y+rnd (15),1;print "MISS";goto 42      +46=1046
62 J=Jxrnd (2);S=S-J;line U,Y,4;line X,Y,1;&(21)=15;gosub 2;gosub 4          +46=1092
64 print "HIT";gosub 2;if S<1print "DESTROYED";goto 5                        +32=1124
66 print "DAMAGE:",#1,S;print #10,@(D);goto 2                               +31=1155

70 S=0;F=0;E=300;&(19)=20;&(20)=10;clear ;NT=3;BC=16                          +47=1202
72 FC=7;for A=21to 40;@(A)=50;next A                                         +27=1229
74 clear ;for I=1to 20;K=rnd (4)-1;@(I)=K;next I;Q=rnd (20);T=rnd (5)        +46=1275
76 CY=3;print #3,K,"KLINGONS";gosub 2;goto 5                                 +30=1305

80 gosub 2                                                                   + 5=1310
                                                                  SZ=          490
   40 @( ) arrays = 80                                            total      +80=1390
```

÷ means div (divide) x means mul (multiply)

This version has 42 lines (84 bytes for line numbers and 42 for the "GO"s) 126 bytes total
original 102 lines (204 bytes for line numbers and 102 for the "GO"s) 306 bytes total 180 bytes saved